Adaptive Numerical Integration to Infinity

Devised and coded by **John Trenholme**
Maple worksheet "AdaptiveNumericalIntegrationToInfinity.mw"
Started 2012-11-29 Changes to 2013-01-08

```
> restart; version = kernelopts( version);
  StringTools:-FormatTime( "It is now %Y-%m-%d %T");
```

$$version = Maple\ 16.02,\ IBM\ INTEL\ NT,\ Nov\ 18\ 2012,\ Build\ ID\ 788210$$

$$\text{"It is now 2013-01-09 19:37:28"} \tag{1}$$

Set numeric precision.

```
> WD := 16: `Working Digits = WD` = WD;
  Digits := WD;
```

$$Working\ Digits\ =\ WD = WD = 16$$

$$Digits := 16 \tag{2}$$

▶ *Default Plot Options*

To Infinity, but not beyond!

Suppose we want to get an approximation to the value of the one-dimensional integral of a reasonably-behaved integrand (no singularities, for example) by means of numerical integration. To control the amount of work needed to get a specified accuracy, we want to use an adaptive method that puts a lot of points where the integrand is large or rapidly varying, and only a few points in small or smooth regions.

This works well if the integration is over a finite interval, but we have a problem with infinite intervals. We need to somehow compress infinity into a finite region. The mathematical giants of yesteryear have come up with a number of effective methods of doing this, all of which involve the introduction of a new integration variable that does the desired mapping of infinity to a finite value.

Consider the basic case of the integral of f(x) from 1 to ∞. We can convert other cases with infinity as one limit to this one by translation, perhaps with a sign change if the shifted integral is from -∞ to -1. Introduce the new variable of integration u(x) with the requirements that it maps the initial infinite interval to a finite interval, and that its inverse x(u) is single-valued (i.e., u(x) is monotonic). Use the Jacobian of the transformation (just a derivative in this one-variable case) to get the identity:

```
> Int( f(x), x = 1 .. infinity) =
    Int( f( x(u)) / Diff( u(x), x), u = u(1) .. u(infinity));
```

$$\int_1^{\infty} f(x)\,\mathrm{d}x = \int_{u(1)}^{u(\infty)} \frac{f(x(u))}{\frac{\mathrm{d}}{\mathrm{d}x} u(x)}\,\mathrm{d}u \tag{3}$$

Since we presume that we have the inverse function x(u), which is needed as the argument to f(x), we can put the derivative of x(u) on top to get the equivalent:

```
> Int( f(x), x = 1 .. infinity) =
    Int( f( x(u)) * Diff( x(u), u), x = u(1) .. u(infinity));
```

$$\int_1^{\infty} f(x)\,\mathrm{d}x = \int_{u(1)}^{u(\infty)} f(x(u)) \left(\frac{\mathrm{d}}{\mathrm{d}u} x(u)\right) \mathrm{d}x \tag{4}$$

The Transformation u = 1 / x

As a simple (and common) example, take u to be 1/x. Then:

```
> u = 1 / x;
  x = solve( %, x);
  Diff( x(u), u) = diff( 1 / u, u);
  plot( 1 / x, x = 1 .. 10, u = 0 .. 1);
```

$$u = \frac{1}{x}$$

$$x = \frac{1}{u}$$

$$\frac{d}{du}\, x(u) = -\frac{1}{u^2}$$

In this case, the limits to the new integral are reversed, and when we put them back into order we cancel the minus sign on the derivative. This gives the final result:

```
> Int( f(x), x = 1 .. infinity) = Int( f( 1 / u) / u^2, u = 0 .. 1)
  ;
```

(5)

$$\int_{1}^{\infty} f(x)\,\mathrm{d}x = \int_{0}^{1} \frac{f\left(\frac{1}{u}\right)}{u^2}\,\mathrm{d}u \tag{5}$$

The tail of the original integral that goes to infinity is mapped to the part of the new integral near 0. Obviously, we have a divide-by-zero problem in both the function argument and the Jacobian at u = 0, so we must handle that point as a special case. Perhaps we avoid zero by a tiny bit (perhaps we start numerical integration at 1E-100 or so), or handling zero as a special case. In the region near u = 0, the integrand, expressed in terms of x, becomes:

```
> limit( subs( u = 1 / x, f( 1 / u) / u^2), x = infinity);
```

$$\lim_{x \to \infty} f(x)\, x^2 \tag{6}$$

We see from this that we should use the 1/x transformation only when $x^2 \cdot f(x)$ is constant or decreasing for large x. If the tail drops off more slowly than x^{-2} near infinity, we will get a singularity (hopefully integrable) in the transformed integral.

As a concrete example, consider the function:

```
> f1 := x -> 2 / (Pi * (1 + (x - 1)^2));
```

$$f1 := x \rightarrow \frac{2}{\pi\left(1 + (x-1)^2\right)} \tag{7}$$

This is just a Lorentzian, shifted so the peak is at x = 1 and scaled for unit area:

```
> plot( f1(x), x = 1 .. 5, integrand = 0 .. evalf( f1(1)));
  `---------- the analytic value of the integral is ----------`;
  Int( f1(x), x = 1 .. infinity) = int( f1(x), x = 1 .. infinity);
```

---------- *the analytic value of the integral is* ----------

$$\int_1^\infty \frac{2}{\pi\left(1+(x-1)^2\right)}\,dx = 1 \tag{8}$$

Define the transformed function that we will integrate over a finite interval. It is smooth and well-behaved, even at u = 0. Carry out a numeric integration of the new function. We get the same result as the analytic integral of the initial function:

```
> g1 := unapply( subs( x = 1 / u, f1(x)) / u^2, u):
  g1 = simplify( g1( u));
  g1 = 2 / (Pi * (u^2 + (1 - u)^2)); # rewrite for clarity
  plot( g1(u), u = 0 .. 1, integrand);
  integral = evalf( Int( g1(u), u = 0 .. 1));
```

$$g1 = \frac{2}{\pi\left(2\,u^2+1-2\,u\right)}$$

$$g1 = \frac{2}{\pi\left(u^2+(1-u)^2\right)}$$



$$integral = 1.000000000000000 \tag{9}$$

The general cases with the u = 1/x transformation are (we could set the first upper limit to 1 by using u = a/x, but there's probably no advantage to doing that):

```
> `---------- positive tail with a > 0 ----------`;
  Int( f(x), x = a .. infinity) =
    Int( f(1 / u) / u^2, u = 0 .. 1 / a);
  `---------- negative tail with b < 0 ----------`;
  Int( f(x), x = -infinity .. b) =
    Int( f(1 / u) / u^2, u = 1 / b .. 0);
```

---------- *positive tail with a > 0* ----------

$$\int_a^{\infty} f(x)\,\mathrm{d}x = \int_0^{\frac{1}{a}} \frac{f\left(\frac{1}{u}\right)}{u^2}\,\mathrm{d}u$$

---------- *negative tail with b < 0* ----------

(10)

$$\int_{-\infty}^{b} f(x)\, dx = \int_{\frac{1}{b}}^{0} \frac{f\left(\frac{1}{u}\right)}{u^2}\, du \qquad \textbf{(10)}$$

Handling Tails That Fall Off As $\frac{1}{x^P}$ with $1 \leqslant P \leqslant 2$

The u = 1 / x transformation we used can give trouble, even for integrals that give a finite result. Consider a simple power-law integrand, which has a finite integral for any power P > 1. The transformation we have used introduces a singularity at u = 0 when P is between 1 and 2. The singularity is integrable, but adaptive integrators may not like it.:

```
> f2 := (x, P) -> (P - 1) / x^P;
  integral = int( f2(x, P), x = 1 .. infinity) assuming P > 1;
  g2 := (x, P) -> subs( x = 1 / u, f2(x, P)) / u^2:
  g2 = g2(u, P);
  p_ := 1.01: `---------- for the case P` = p_;
  plot( g2(u, p_), u = 0 .. 1, integrand = 0 .. 1);
  `----- Maple can do this singularity numerically`;
  integral = evalf( Int( f2(x, p_), x = 1 .. infinity));
```

$$f2 := (x, P) \rightarrow \frac{P-1}{x^P}$$

$$integral = 1$$

$$g2 = \frac{P-1}{\left(\frac{1}{u}\right)^P u^2}$$

$$\text{----------}\ for\ the\ case\ P = 1.01$$



$$\text{----- } \textit{Maple can do this singularity numerically}$$

$$integral = 1. \quad \textbf{(11)}$$

To tame power-law behavior that goes as x^P as x becomes large, we can adjust the power on the transformation. Let the exponent be Q. We previously used Q = 1, but in general we have:

```
> u = 1 / x^Q;
  x = 1 / u^(1 / Q);
  Diff( x(u), u) = diff( 1 / u^(1 / Q), u);
```

$$u = \frac{1}{x^Q}$$

$$x = \frac{1}{u^{\frac{1}{Q}}}$$

$$\frac{d}{du}\, x(u) = -\frac{1}{u^{\frac{1}{Q}}\, Q\, u} \quad \textbf{(12)}$$

This will keep the transformed integrand finite if Q ⩽ P-1. Clearly we can do this only if P > 1, as might be expected since only then is the integral finite. For the unit-area power-law function f2(x) above with P = 1.01, we need to have Q ⩽ 0.01. When we plot the integrand for a few values of Q, we see that 0.01 gives a nice flat result, and 0.009 decreases near u = 0, but that Q = 0.011 is too large and has the undesired singular behavior:

```
> g3 := (u, P, Q) -> (P - 1) / (Q * u^(1 + (1 - P) / Q));
  q_ := 0.01:
  plot( [seq( g3(u, p_, q_), q_ = [0.009, 0.01, 0.011])],
    u = 0 .. 1, integrand = 0 .. 2);
```

$$g3 := (u, P, Q) \to \frac{P-1}{Q\,u^{1+\frac{1-P}{Q}}}$$



One simple case is Q = 1/2, which gives us a transformation that will handle functions that drop off as $x^{-1.5}$ or faster, instead of the x^{-2} we could handle with u = 1/x.

```
> u = 1 / sqrt(x);
  x = 1 / u^2;
  Diff( x(u), u) = diff( 1 / u^2, u);
  Int( f(x), x = 1 .. infinity) =
    Int( 2 * f( 1 / u^2) / u^3, u = 0 .. 1);
```

$$u = \frac{1}{\sqrt{x}}$$

$$x = \frac{1}{u^2}$$

$$\frac{d}{du}x(u) = -\frac{2}{u^3}$$

$$\int_1^{\infty} f(x)\,dx = \int_0^1 \frac{2 f\left(\frac{1}{u^2}\right)}{u^3}\,du \tag{13}$$

We easily proceed to $x = \frac{1}{u^4}$ and handle tails going as $x^{-1.25}$ or faster:

```
> u = 1 / sqrt( sqrt(x));
  x = 1 / u^4;
  Diff( x(u), u) = diff( 1 / u^4, u);
  Int( f(x), x = 1 .. infinity) =
    Int( 4 * f( 1 / u^4) / u^5, u = 0 .. 1);
```

$$u = \frac{1}{x^{1/4}}$$

$$x = \frac{1}{u^4}$$

$$\frac{d}{du}x(u) = -\frac{4}{u^5}$$

$$\int_1^{\infty} f(x)\,dx = \int_0^1 \frac{4 f\left(\frac{1}{u^4}\right)}{u^5}\,du \tag{14}$$

Continuing with integer powers, the transformation to handle tails going as $x^{-\frac{(N+1)}{N}}$ or faster is:

```
> Int( f(x), x = 1 .. infinity) =
    Int( N * f( 1 / u^N) / u^(N + 1), u = 0 .. 1);
```

$$\int_1^{\infty} f(x)\,dx = \int_0^1 \frac{N f\left(\frac{1}{u^N}\right)}{u^{N+1}}\,du \tag{15}$$

Another Transformation for Power-Law Tails with P $\geqslant$ 2

For integrands that fall of as $\frac{1}{x^2}$ or faster, we can use a transformation that maps the interval 0 to ∞ to the interval from 0 to 1 and is monotone upward. This is essentially the same as the simple power law we used previously, but with ∞ at u = 1 and the ability to start at x = 0 with no trouble. We have:

```
> u = x / (1 + x);
  x = u / (1 - u);
  Diff( x(u), u) = simplify( diff( u / (1 - u), u));
  plot( x / (1 + x), x = 0 .. 10, u = 0 .. 1);
```

$$u = \frac{x}{1+x}$$

$$x = \frac{u}{1-u}$$

$$\frac{d}{du}\, x(u) = \frac{1}{(-1+u)^2}$$



Now our initial integral starts at x = 0, which may be easier to translate to. We get:

```
> Int( f(x), x = 0 .. infinity) =
    Int( f( u / (1 - u)) / (1 - u)^2, u = 0 .. 1);
```

$$\int_0^{\infty} f(x)\, dx = \int_0^1 \frac{f\left(\frac{u}{1-u}\right)}{(1-u)^2}\, du \tag{16}$$

The requirement on the behavior of the tails of f(x) is essentially the same as in the u = 1/x case:

```
> limit( subs( u = x / (1 + x), f( u / (1 - u)) / (1 - u)^2),
    x = infinity);
  simplify(%);
```

$$\lim_{x\to\infty} \frac{f\left(\frac{x}{(1+x)\left(1-\frac{x}{1+x}\right)}\right)}{\left(1-\frac{x}{1+x}\right)^2}$$

$$\lim_{x\to\infty} f(x)\,(1+x)^2 \qquad \textbf{(17)}$$

There is a divide-by-zero problem at u = 1, both for the argument to the function and for the integrand as a whole. We can stop before 1, although we cannot get as close as in the previous transformation (the minimum offset in IEEE 754 arithmetic is 1.1E-16). A better method is to adjust the transformation so that u = 1 produces a large but not infinite value. The largest IEEE 754 number smaller than 1 is 1 - 1.1E-16, so we do the following:

```
> S := 1 - 1.1E-16;
  x = u / (1 - 'S' * u); # now this is finite, but large, at u = 1
  Diff( x(u), u) = simplify( diff( u / (1 - 'S' * u), u));
```

$$S := 0.9999999999999999$$

$$x = \frac{u}{1 - S\,u}$$

$$\frac{d}{du}\,x(u) = \frac{1}{(-1 + S\,u)^2} \qquad \textbf{(18)}$$

The integrand for an unshifted unit-area half Lorentzian is almost the same as in the previous case, and the numerical result is the same to 1E-16 relative accuracy:

```
> f4 := x -> 2 / (Pi * (1 + x^2));
  g4 := unapply( subs( x = u / (1 - S * u), f4(x)) / (1 - S * u)^2,
  u):
  g4 = g4( u);
  g4 = 2 / (Pi * (u^2 + (1 - 'S' * u)^2));
  plot( g4( u), u = 0 .. 1);
  integral = evalf( Int( g4(u), u = 0 .. 1));
```

$$f4 := x \to \frac{2}{\pi\,(x^2+1)}$$

$$g4 = \frac{2}{\pi\left(\frac{u^2}{(1 - 0.9999999999999999\,u)^2} + 1\right)(1 - 0.9999999999999999\,u)^2}$$

$$g4 = \frac{2}{\pi\,(u^2 + (1 - S\,u)^2)}$$



$$integral = 0.9999999999999999 \tag{19}$$

A Transformation for Exponential Tails

We can handle more rapid fall-offs than power-law by different transformations. For example, consider an exponentially-dropping tail. Use an exponential transformation with dropoff-rate coefficient A to transform the interval from 0 to ∞ to the interval from 0 to 1:

```
> u = exp( -A * x);
  x = -ln( u) / A;
  Diff( x(u), u) = diff( -ln( u) / A, u);
```

$$u = e^{-A x}$$

$$x = -\frac{\ln(u)}{A}$$

$$\frac{d}{du} x(u) = -\frac{1}{u A} \tag{20}$$

The transformed integral has the form:

```
> Int( f(x), x = 0 .. infinity) =
    Int( f( -ln( u) / A) / (A * u), u = 0 .. 1);
```

$$\int_0^{\infty} f(x)\,\mathrm{d}x = \int_0^1 \frac{f\left(-\frac{\ln(u)}{A}\right)}{A\,u}\,\mathrm{d}u \qquad \textbf{(21)}$$

Once again we must be cautious at u = 0, and avoid that value or handle it as a special case.

Consider a sech function, which has exponential tails by definition. Adjust the rate of tail falloff by an adjustable coefficient C:

```
> f5 := (x, C) -> 2 * C * sech(C * x) / Pi;
  integral = int( f5(x, C), x = 0 .. infinity) assuming C > 0;
```

$$f5 := (x, C) \rightarrow \frac{2\,C\,\mathrm{sech}(C\,x)}{\pi}$$

$$integral = 1 \qquad \textbf{(22)}$$

With A = 1, we must have C ⩾ 1 to avoid singular behavior near u = 0, and Maple is unhappy if the tail exponent is not tamed by the transformation. In all cases, we need to reduce the requested accuracy a to avoid recursion errors, with a larger reduction when there is a singularity. The exact procedure to use depends on how well your numerical integrator handles singular integrands; the Maple integrator is pretty good :

```
> g5 := (x, A, C) -> subs( x = -ln( u) / A, f5(x, C)) / (A * u):
  g5 = g5( u, A, C);
  plot( [ seq( g5( u, 1, c_), c_ = [0.5, 1, 2])], u = 0 .. 1);
  `----- case for C = 0.5`;
  evalf( Int( g5(u, 1, 0.5), u = 0 .. 1), WD - 5);
  `----- case for C = 1`;
  integral = evalf( Int( g5(u, 1, 1), u = 0 .. 1), WD - 1);
  `----- case for C = 2`;
  integral = evalf( Int( g5(u, 1, 2), u = 0 .. 1), WD - 1);
```

$$g5 = \frac{2\,C\,\mathrm{sech}\left(-\frac{C\ln(u)}{A}\right)}{\pi\,A\,u}$$



----- case for C = 0.5

1.0000000000

----- case for C = 1

integral = 1.00000000000000

----- case for C = 2

integral = 0.999999999999998 **(23)**

Setting A = 0.5, cases for C = 0.5, 1, 2 all behave well, although C = 2 is getting a bit bumpy near u = 1. We conclude that the exponential falloff in the transformation needs to be slower than the falloff in the integrand, but not too much smaller. That is to say, if the tail falls off as exp(-C*x) then we want to transform to a variable u = exp(-A*x) with A ≤ C but not too much smaller:

```
> plot( [ seq( g5( u, 0.5, c_), c_ = [0.5, 1, 2])], u = 0 .. 1);
  `----- case for C = 0.5`;
  integral = evalf( Int( g5(u, 0.5, 0.5), u = 0 .. 1), WD - 1);
  `----- case for C = 1`;
  integral = evalf( Int( g5(u, 0.5, 1), u = 0 .. 1), WD - 1);
  `----- case for C = 2`;
  integral = evalf( Int( g5(u, 0.5, 2), u = 0 .. 1), WD - 1);
```



----- case for C = 0.5

$integral = 1.00000000000000$

----- case for C = 1

$integral = 0.999999999999998$

----- case for C = 2

$integral = 1.00000000000000$ **(24)**

We could just as well have used one of our power-law transformations to do the exponential case. For example, using the u = x/(1+x) transformation with the sech function gives integrands that drop off sharply near u = 1, but are still reasonably smooth and easily integrable with an adaptive method. We still have to drop the precision by a digit to get Maple to do the job:

```
> g6 :=
    unapply( subs( x = u / (1 - S * u), f5(x, C)) / (1 - S * u)^2,
  u, C):
  g6 = g6( u, C);
  plot( [ seq(g6(u, c_), c_ = [0.5, 1, 2])], u = 0 .. 1);
  `----- case for C = 0.5`;
  integral = evalf( Int( g6(u, 0.5), u = 0 .. 1), WD - 1);
  `----- case for C = 1`;
  integral = evalf( Int( g6(u, 1), u = 0 .. 1), WD - 1);
  `----- case for C = 2`;
  integral = evalf( Int( g6(u, 2), u = 0 .. 1), WD - 1);
```

$$g6 = \frac{2\,C\,\mathrm{sech}\left(\frac{C\,u}{1 - 0.9999999999999999\,u}\right)}{\pi\,(1 - 0.9999999999999999\,u)^2}$$



----- case for C = 0.5

$integral = 1.00000000000000$

----- case for C = 1

$integral = 0.999999999999999$

----- case for C = 2

$integral = 0.999999999999998$ **(25)**

```
> #~~~~~~~~~~~~~~~~~~~~~~~~ end of file ~~~~~~~~~~~~~~~~~~~~~~~~
```